



PascalABC.NET
система программирования



Подготовка к ЕГЭ по информатике: решение типовых задач на PascalABC.NET

Ячmeneва Наталья Николаевна

третья Всероссийская научно-методическая конференция «Использование
системы программирования PascalABC.NET в обучении программированию»

Южный федеральный университет

13 ноября 2021 год

Подготовка к ЕГЭ по информатике: решение типовых задач на PascalABC.NET

Типовые задачи:

- 6, 22
- 14, 16, 17, 24, 25, 26, 27

Подготовка к ЕГЭ по информатике: решение задачи 6 на PascalABC.NET

Определите, при каком
наибольшем введённом значении
переменной *s* программа выведет
число 64. Для Вашего
удобства программа представлена
на четырёх языках
программирования.

Паскаль

```
var s, n: integer;  
begin  
  readln(s);  
  s := s div 10;  
  n := 1;  
  while s < 51 do  
  begin  
    s := s + 5;  
    n := n * 2  
  end;  
  writeln(n)  
end.
```

Подготовка к ЕГЭ по информатике: решение задачи 6 на PascalABC.NET

```
1  var
2    s, n: integer;
3
4  begin
5    readln(s);
6    s := s div 10;
7    n := 1;
8    while s < 51 do
9    begin
10       s := s + 5;
11       n := n * 2
12    end;
13    writeln(n)
14 end.
```

```
//var
// s, n: integer;

begin
    // readln (s);
    for var i := 1 to 520 do
    begin
        var s := i;
        s := s div 10;
        var n := 1;
        while s < 51 do
        begin
            s := s + 5;
            n := n * 2
        end;
        //writeln(n)
        if n = 64 then
            print(i);
        end;
    end.
```

Подготовка к ЕГЭ по информатике: решение задачи 6 на PascalABC.NET

```
1  ##
2  for var i := 1 to 520 do
3  begin
4      var s := i;
5      s := s div 10;
6      var n := 1;
7      while s < 51 do
8      begin
9          s := s + 5;
10         n := n * 2
11     end;
12     if n = 64 then
13         print(i);
14 end;
```

Окно вывода

⏏ ×

```
210 211 212 213 214 215 216 217 218 219 220 221 222 223 224 225 226 227 228 229 230 231
232 233 234 235 236 237 238 239 240 241 242 243 244 245 246 247 248 249 250 251 252 253
254 255 256 257 258 259
```

Подготовка к ЕГЭ по информатике: решение задачи 22 на PascalABC.NET

Ниже на четырёх языках
программирования записан
алгоритм. Получив на
вход число x , этот алгоритм
печатает два числа: L и M .
Укажите наибольшее
число x , при вводе которого
алгоритм печатает сначала 4,
потом 5.

Паскаль

```
var x, L, M, Q: integer;  
begin  
  readln(x);  
  Q := 9;  
  L := 0;  
  while x >= Q do  
  begin  
    L := L + 1;  
    x := x - Q;  
  end;  
  M := x;  
  if M < L then  
  begin  
    M := L;  
    L := x;  
  end;  
  writeln(L);  
  writeln(M);  
end.
```

Подготовка к ЕГЭ по информатике: решение задачи 22 на PascalABC.NET

```
var
  x, L, M, Q: integer;
|
begin
  //readln(x);
  for var y := 1 to 10000 do
  begin
    x := y;
    Q := 9;
    L := 0;
    while x >= Q do
    begin
      L := L + 1;
      x := x - Q;
    end;
    M := x;
    if M < L then begin
      M := L;
      L := x;
    end;
    //writeln(L);
    //writeln(M);
    if (L = 4) and (M = 5) then
      println(y);
  end;
end.
```

```
1  ##|
2  for var y := 1 to 10000 do
3  begin
4    var(x, Q, L) := (y, 9, 0);
5    while x >= Q do
6      (L, x) := (L + 1, x - Q);
7    var M := x;
8    if M < L then
9      (M, L) := (L, x);
10   if (L = 4) and (M = 5) then
11     println(y);
12 end;
```

Окно вывода

41
49

Подготовка к ЕГЭ по информатике: решение задачи 14 на PascalABC.NET

Значение арифметического выражения

$$3 \cdot 4^{38} + 2 \cdot 4^{23} + 4^{20} + 3 \cdot 4^5 + 2 \cdot 4^4 + 1$$

записали в системе счисления с основанием 16. Сколько значащих нулей содержится в этой записи?



```
1  ##
2  var n := 3 * 4bi ** 38 + 2 * 4bi ** 23 + 4bi ** 20 + 3 * 4bi ** 5 + 2 * 4bi ** 4 + 1;
3  var count0 := 0;
4  while n <> 0 do
5  begin
6      if n mod 16 = 0 then
7          count0 += 1;
8      n := n div 16;
9  end;
10 print(count0); |
```

Окно вывода

15

Подготовка к ЕГЭ по информатике: решение задачи 16 на PascalABC.NET

Алгоритм вычисления значения функции $F(n)$, где n – натуральное число, задан следующими соотношениями:

$F(n) = 1$ при $n = 1$;

$F(n) = n + F(n - 1)$, если n чётно,

$F(n) = 2 \times F(n - 2)$, если $n > 1$ и при этом n нечётно.

Чему равно значение функции $F(26)$?

```
9  function f(n: integer): integer;
10 begin
11     if n = 1 then
12         Result := 1
13     else if n mod 2 = 0 then
14         Result := n + f(n - 1)
15     else
16         Result := 2 * f(n - 2);
17 end;
18
19 function f1(n: integer): integer :=
20     n = 1 ? 1 : (n mod 2 = 0 ? n + f1(n - 1) : 2 * f1(n - 2));
21
22 begin
23     print(f(26));
24     print(f1(26));
25 end.
```

{CASH}

Окно вывода

4122 4122

Подготовка к ЕГЭ по информатике: решение задачи 17 на PascalABC.NET

В файле содержится последовательность целых чисел. Элементы последовательности могут принимать целые значения от $-10\,000$ до $10\,000$ включительно. Определите количество пар последовательности, в которых хотя бы одно число делится на 3, а сумма элементов пары не более максимального элемента последовательности, кратного 3. В ответе запишите количество найденных пар, затем максимальную из сумм элементов таких пар. В данной задаче под парой подразумевается два идущих подряд

```
32 // с массивами
33 var a := ReadLines('17.txt').Select(x -> x.ToInteger).ToArray;
34 var (maxSum, count) := (Integer.MinValue, 0);
35 for var i := 0 to a.Length - 2 do
36 begin
37     if (a[i] mod 3 = 0) or (a[i + 1] mod 3 = 0) then
38     begin
39         count += 1;
40         if a[i] + a[i + 1] > maxSum then
41             maxSum := a[i] + a[i + 1];
42     end;
43 end;
44 Print(count, maxSum);
```

Ответ:

--	--

Окно вывода

2802 1990

Подготовка к ЕГЭ по информатике: решение задачи 17 на PascalABC.NET

Рассматривается множество целых чисел, принадлежащих числовому отрезку $[1016; 7937]$, которые делятся на 3 и не делятся на 7, 17, 19, 27.

Найдите количество таких чисел и максимальное из них.

В ответе запишите два целых числа без пробелов и других дополнительных символов: сначала количество, затем максимальное число.

Для выполнения этого задания можно написать программу или воспользоваться редактором электронных таблиц.

Источник: Демонстрационная версия ЕГЭ—2021 по информатике.

```
1 ##  
5 var a := Arr(1016..7937);  
5 var b := a.Where(x -> x mod 3 = 0).Where(x -> not x.DivsAny(7, 17, 19, 27));  
7 Print(b.Count, b.Max);  
}
```

Окно вывода

1568 7935

Подготовка к ЕГЭ по информатике: решение задачи 24 на PascalABC.NET

Текстовый файл состоит из символов P, Q, R и S.

Определите максимальное количество идущих подряд символов в прилагаемом файле, среди которых нет идущих подряд символов P.

Для выполнения этого задания следует написать программу.

```
29 var f := OpenRead('24.txt'); //Assign(input, '24.txt')
30 var (maxc, k, x) := (0, 1, f.ReadChar);
31 while not f.Eof do
32 begin
33     var y := f.ReadChar;
34     if (x = 'P') and (y = 'P') then
35     begin
36         if k > maxc then
37             maxc := k;
38         k := 1;
39     end
40     else
41         k += 1;
42     x := y;
43 end;
44 Print(maxc);
```

```
10 ##
11 var f := ReadLines('24.txt').JoinToString;
12 var (maxc, i) := (0, -1);
13 while true do
14 begin
15     var j := f.indexOf('PP', i + 1);
16     if j = -1 then break;
17     if j - i > maxc then
18         maxc := j - i;
19     i := j;
20 end;
21 print(maxc);
```

Окно вывода

188

Подготовка к ЕГЭ по информатике: решение задачи 25 на PascalABC.NET

Пусть M – сумма минимального и максимального натуральных делителей целого числа, не считая единицы и самого числа. Если таких делителей у числа нет, то значение M считается равным нулю.

Напишите программу, которая перебирает целые числа, большие 700 000, в порядке возрастания и ищет среди них такие, для которых значение M оканчивается на 8. Выведите первые пять найденных чисел и соответствующие им значения M .

Формат вывода: для каждого из пяти таких найденных чисел в отдельной строке сначала выводится само число, затем – значение M .

Строки выводятся в порядке возрастания найденных чисел.

Окно вывода

700005	233338
700007	100008
700012	350008
700015	140008
700031	24168

```
13 ##
14 var (c,n) := (0,700000);
15 while c < 5 do
16 begin
17     n += 1;
18     var a := Arr(2..(n div 2)).Where(x -> n mod x = 0).ToArray;
19     if (a.Count > 1) and ((a[0] + a[^1]) mod 10 = 8) then
20 begin
21     Println(n, ' ', a[0] + a[^1]);
22     c += 1;
23 end;
24 end;
```

Подготовка к ЕГЭ по информатике: решение задачи 25 на PascalABC.NET

Напишите программу, которая ищет среди целых чисел, принадлежащих числовому отрезку [174457; 174505], числа, имеющие ровно два различных натуральных делителя, не считая единицы и самого числа. Для каждого найденного числа запишите эти два делителя в две соседних столбца на экране с новой строки в порядке возрастания произведения этих двух делителей. Делители в строке также должны следовать в порядке возрастания.

Источник: Демонстрационная версия ЕГЭ—2021 по информатике.

```
20 var c := Arr(174457..174505);
21 foreach var n in c do
22 begin
23     var a := Arr(2..(n div 2)).Where(x -> n mod x = 0).ToArray;
24     if a.Count = 2 then
25         Println(a[0], ' ', a[1]);
26 end;
```

Окно вывода	
3	58153
7	24923
59	2957
13	13421
149	1171
5	34897
211	827
2	87251

Подготовка к ЕГЭ по информатике

решение задачи 26

26

Системный администратор раз в неделю создаёт архив пользовательских файлов. Однако объём диска, куда он помещает архив, может быть меньше, чем суммарный объём архивируемых файлов.

Известно, какой объём занимает файл каждого пользователя.

По заданной информации об объёме файлов пользователей и свободном объёме на архивном диске определите максимальное число пользователей, чьи файлы можно сохранить в архиве, а также максимальный размер имеющегося файла, который может быть сохранён в архиве, при условии, что сохранены файлы максимально возможного числа пользователей.

Входные данные.

В первой строке входного файла находятся два числа: S – размер свободного места на диске (натуральное число, не превышающее 10 000) и N – количество пользователей (натуральное число, не превышающее 1000). В следующих N строках находятся значения объёмов файлов каждого пользователя (все числа натуральные, не превышающие 100), каждое – в отдельной строке.

Запишите в ответе два числа: сначала наибольшее число пользователей, чьи файлы могут быть помещены в архив, затем – максимальный размер имеющегося файла, который может быть сохранён в архиве, при условии, что сохранены файлы максимально возможного числа пользователей.

Пример входного файла:

100 4

80

30

50

40

Ответ:

--	--

Подготовка к ЕГЭ по информатике: решение задачи 26 на PascalABC.NET

```
40 ##
41 assign(input, '26_demo.txt');
42 var (V, N) := ReadInteger2;
43 var a := ReadArrInteger(N);
44 a.Sort;
45 var (s, imax) := (0, -1);
46 for var i := 0 to a.Length - 1 do
47     if s + a[i] <= V then
48         (s, imax) := (s + a[i], i);
49 print(imax + 1);
50 s -= a[imax];
51 var c := a.FindLast(x -> x + s <= V);
52 println(c);
53
```

Окно вывода

568 50

Подготовка к ЕГЭ по информатике

решение задачи 27

27

Дана последовательность из N натуральных чисел. Рассматриваются все её непрерывные подпоследовательности, такие что сумма элементов каждой из них кратна $k = 43$. Найдите среди них подпоследовательность с максимальной суммой, определите её длину. Если таких подпоследовательностей найдено несколько, в ответе укажите количество элементов самой короткой из них.

Входные данные

Даны два входных файла (файл A и файл B), каждый из которых содержит в первой строке количество чисел N ($1 \leq N \leq 10\,000\,000$). Каждая из следующих N строк содержит одно натуральное число, не превышающее 10 000.

Пример организации исходных данных во входном файле:

7

1

3

4

93

8

5

95

В ответе укажите два числа: сначала значение искомой длины для файла A , затем – для файла B .

Предупреждение: для обработки файла B не следует использовать переборный алгоритм, вычисляющий сумму для всех возможных вариантов, поскольку написанная по такому алгоритму программа будет выполняться слишком долго.

Ответ:

Подготовка к ЕГЭ по информатике: решение задачи 27 на PascalABC.NET

```
33 var
34   k := 43;
35
36 procedure A(fname: string);
37 begin
38   // var f := (ReadLines(fname).Select(x -> x.ToInteger).ToArray)[1:];
39   assign(input, fname);
40   var N := readInteger;
41   var f := readArrInteger(N);
42   var (max, c) := (0, 0);
43   for var i := 0 to f.Length - 2 do
44     for var j := f.Length downto i + 1 do
45       begin
46         var a := f[i:j];
47         var s := a.sum;
48         if (s mod k = 0) then
49           if (s > max) then
50             (max, c) := (s, a.count)
51           else if (s = max) and (a.count < c) then
52             c := a.count;
53       end;
54   Println(c);
55 end;
88 begin
89   A('27_A.txt');
90   B('27_A.txt'); // для проверки правильности процедуры
91
92   B('27_B.txt');
93 end.
```

```
57 procedure B(fname: string);
58 begin
59   assign(input, fname);
60   var n: longint := ReadInteger;
61   var mins := Arr(Integer.MaxValue) * k;
62   var minl := Arr(0) * k;
63   var (max, c, sum) := (0, 0, 0);
64   for var i := 1 to n do
65     begin
66       var x := ReadlnInteger;
67       sum += x;
68       var d := sum mod k;
69       if d = 0 then
70         (max, c) := (sum, i)
71       else
72         begin
73           var (ms, l) := (sum - mins[d], i - minl[d]);
74           if ms > max then
75             (max, c) := (ms, l);
76           if (ms = max) and (l < c) then
77             c := l;
78         end;
79       if sum < mins[d] then
80         (mins[d], minl[d]) := (sum, i);
81     end;
82   Println(c);
83 end;
```

Окно вывода

185
185
329329